
1. Что такое DLL
Под давлением все ухудшается.
 Из законов Мэрфи

При разработке программ часто оказывается, что разным программным приложениям требуются одни и те же объекты, их свойства методы, процедуры и функции. Например, почти все программы выводят информацию на экран и пользуются стандартными объектами интерфейса Windows (окна, кнопки, меню…) Было бы в высшей степени неразумно запихивать код отрисовки каждого такого элемента во все программы. 

Таким образом, возникает задача разделения большой программы на отдельные независимые модули, каждый из которых содержит определенный набор процедур и функций. Процедуры и функции такого модуля может вызывать любая другая программа. Подобные модули получили название динамически подключаемых библиотек (DLL – dynamically linked library, Рис.  1). Слово "динамический" указывает на то, что подключение библиотеки происходит не на этапе компиляции, а уже после запуска программы. 

Рис.  1 – DLL – библиотеки.

DLL-библиотеки нашли самое широкое применение в большинстве программ. Скажем, сама операционная система Windows включает в свой состав несколько сотен DLL, и заключенная в них функциональность может использоваться нашими программами во избежание очередного изобретения велосипеда. 

В каких случаях следует применять DLL? Во-первых, если вы разрабатываете несколько независимых программ и в них используются одни и те же фрагменты программного кода. Тогда такие фрагменты следует вынести в библиотеку. Это не только экономит место на диске, но и облегчает внесение изменений в проект – придется править только одну DLL, а не много exe-программ. Во-вторых, если вы хотите дать возможность пользователям вашей гениальной программы применять ее в своих разработках. В этом случае программу обычно разделяют на маленький exe-файл и большую DLL-библиотеку. Другие разработчики смогут использовать вашу DLL-библиотеку при условии, что вы предоставите описание содержащихся в ней процедур и функций и их параметров. Принято говорить, что процедуры и функции DLL-библиотеки экспортируются в другие программы.
2. Создание DLL-библиотеки на Delphi 

-Как пройти в библиотеку?

Вопрос, заданный на улице в 3 часа ночи.

В среде программирования Delphi предусмотрены встроенные средства для быстрого создания DLL-библиотек. Давайте для определенности создадим библиотеку, содержащую функцию GetArea(a, b, c: REAL):REAL. Данной функции на вход подаются длины сторон треугольника. Функция возвращает площадь заданного треугольника:
FUNCTION GetArea(a, b, c:REAL):REAL;

VAR p:REAL;

BEGIN

 p:=(a+b+c)/2;

 Result:=SQRT(p*(p-a)*(p-b)*(p-c))
END;

Запускаем Delphi, а дальше действуем нетрадиционно. Выбираем пункты меню File ( New ( Other, в открывшемся окошке на закладке New щелкаем по пиктограмме DLL Wizard [image: image2.png]%



. При этом создается файл заготовки DLL-библиотеки. Он очень похож на обычный модуль (unit) Delphi, только начинается с оператора Library. Сохраните проект под именем, которое в будущем получит DLL-библиотека, скажем, GetA. Название GetArea использовать нельзя – оно уже занято именем функции. 
Теперь после оператора USES пишем текст нашей функции, но с некоторыми изменениями в заголовке: 
FUNCTION GetArea(a, b, c:REAL):REAL; export; stdcall;
Ключевое слово EXPORT указывает на то, что данная функция является экспортируемой и будет видна из внешних программ. Слово STDCALL (standard call – стандартная модель вызова) указывает на применяемый протокол передачи параметров в функцию. Стандартная модель вызова совместима с программами, написанными на других языках, в частности, на С, поэтому нашу DLL смогут использовать и программисты-извращенцы, использующие С. 

После текста функции припишем

EXPORTS

 GetArea;

В операторе EXPORTS перечисляются все процедуры и функции, экспортируемые из библиотеки. Это своего рода каталог нашей библиотеки. 

Запустить на выполнение библиотеку невозможно, ее можно только откомпилировать. Для этого выполним пункт меню Project ( Build. Если все было сделано правильно, на диске в текущей директории будет создан файл с именем geta.dll. Это и есть наша библиотека. 

Важное замечание: существует определенная тонкость при передаче процедурам и функциям, находящимся в библиотеке, параметров типа STRING. Для того, чтобы можно было передавать параметры типа STRING, придется в операторы USES и библиотеки, и вызывающей ее программы прописать подключение модуля ShareMem, да еще и обязательно так, чтобы этот модуль шел первым в списке. Мало того, вместе с библиотекой придется обязательно прилагать файл borlndmm.dll (он входит в поставку Delphi). Избежать таких хлопот очень просто: следует для параметров текстового типа использовать типы данных ShortString (фактически обычная строка, но длиной до 255 символов) и PChar (указатель на текстовую строку). К сожалению, в DLL-библиотеки практически невозможно передавать массивы, как статические, так и динамические.
3. Вызов DLL

Нельзя вызвать дождь, начав мыть машину.
Из законов Мэрфи
Существуют два способа вызова процедур и функций из DLL. В первом случае мы заранее, на этапе разработки программы, знаем, какую DLL будем к ней подключать (чаще всего мы сами и создаем эту DLL). Во втором случае мы подключаемся к произвольной библиотеке, в том числе и "чужой". 

3.1.  Статическое связывание

Для реализации первого способа, называемого статическим  связыванием, создайте новое обычное приложение, поместите на форму три поля ввода LabeledEdit1…LabeledEdit3, кнопку и компонент Tlabel. После оператора IMPLEMENTATION добавьте строчку, обеспечивающую импорт функции GetArea из библиотеки geta.dll:

Function GetArea(a,b,c:real):REAL; FAR; EXTERNAL 'geta';
Слово EXTERNAL указывает на то, что тело данной функции находится в библиотеке с указанным именем, а слово FAR задает применение "длинных" четырехбайтных адресов, что необходимо, так как вызывающая программа находится на одной странице памяти, а DLL-библиотека – на другой. Разумеется, файл geta.dll надо поместить в ту же директорию, где находятся все файлы текущего приложения. 
В обработчике нажатия кнопки надо создать массив и передать его библиотечный функции, а результат вывести на экран: 
procedure TForm1.Button1Click(Sender: TObject);

VAR a,b,c:REAL;

begin

 Label1.Caption:='';

 TRY

  a:=StrToFloat(LabeledEdit1.Text);

  b:=StrToFloat(LabeledEdit2.Text);

  c:=StrToFloat(LabeledEdit3.Text)

 EXCEPT

  Label1.Caption:='Ошибка ввода!';

  Exit

 END;

 Label1.Caption:='S='+FloatToStrF(GetArea(a,b,c),ffFixed,10,4)

end;

Запускаем программу. Если все было сделано правильно, то произойдет вызов функции, физически расположенной в DLL-библиотеке, и получение возвращаемого ею значения. 
3.2.  Динамическое связывание

Второй способ вызова называется динамическим связыванием и отличается прежде всего тем, что библиотека помещается в память только на время выполнения, а потом выгружается. Рассмотрим следующий код:
procedure TForm1.Button1Click(Sender: TObject);

TYPE

  TFunc = Function (a,b,c:real):REAL; STDCALL;

CONST DLLName='geta.DLL'; // имя DLL
VAR a,b,c:REAL;

    Handle:THandle;

    F:TFunc;

begin

 Label1.Caption:='';

 TRY

  a:=StrToFloat(LabeledEdit1.Text);

  b:=StrToFloat(LabeledEdit2.Text);

  c:=StrToFloat(LabeledEdit3.Text)

 EXCEPT

  Label1.Caption:='Ошибка ввода!';

  Exit

 END;

 Handle := LoadLibrary(DLLName); // загрузка библиотеки
 IF ( Handle <> 0 ) THEN

  BEGIN
   // получение адреса функции GetArea в библиотеке
   @F:=GetProcAddress(Handle,'GetArea');

   IF Assigned(F) THEN

    BEGIN
     // вызов функции 
     Label1.Caption:='S='+FloatToStrF(F(a,b,c),ffFixed,10,4);
     // выгрузка библиотеки из памяти
     FreeLibrary(Handle)

    END 

  END

 ELSE

  MessageDlg('Не найдена библиотека '+DLLName,mtWarning,[mbOK],0)

end;

Для динамического вызова прежде всего надо загрузить библиотеку при помощи функции LoadLibrary(имя_библиотеки). Эта функция возвращает ссылку на библиотеку, которую мы записываем в переменную типа Thandle.  Теперь надо получить адрес функции GetArea, спрятанной в библиотеке. Его возвращает функция GetProcAddress(ссылка_на_библиотеку,имя_функции). А далее начинаются трюки. Для вызова функции из библиотеки надо описать особый тип данных – ссылку на функцию. В нем указывается, какие аргументы подаются на вход функции и значение какого типа она возвращает. Тип TFunc = Function (a,b,c:real):REAL; STDCALL; соответствует функции с любым именем, на вход которой надо подать три вещественных числа и которая возвращает одно вещественное число, используя стандартную модель вызова. Объявим переменную F типа TFunc. В нее и запишем адрес функции GetArea. Для этого используется символ @, означающий получение адреса переменной. Таким образом, в строке
@F:=GetProcAddress(Handle,'GetArea');
происходит следующее: переменная F размещается в памяти по тому же адресу, по которому в памяти расположен код функции GetArea. После этого переменная F превращается в ссылку на функцию, поэтому вполне законна запись F(a,b,c), означающая вызов функции GetArea с аргументами a,b,c. Наконец, команда FreeLibrary(ссылка_на_библиотеку) выгружает DLL из памяти. 
В приведенном фрагменте предусмотрен ряд "затычек". Проверка условия Handle<>0 позволяет установить, удалось ли загрузить библиотеку (вдруг ее просто нет?)  Условие Assigned(F) проверяет,  удалось ли получить адрес функции в библиотеке.

Динамическое связывание позволяет подключаться к чужим DLL – достаточно знать, какие процедуры и функции в них находятся и с какими параметрами их надо вызывать. 
4. Отладка проектов с DLL
 Если отладка - процесс удаления ошибок, то
 программирование должно быть процессом их внесения. 

Э. Дейкстра 
Так как саму по себе DLL запустить на выполнение невозможно, возникает трудность с отладкой библиотек: в них не удается поставить точки прерывания, отследить значения переменных и т.д. Для отладки надо каким-то образом связать вместе два проекта: саму DLL и вызывающую ее программу. Подобное связывание в Delphi предусмотрено и называется группой проектов (project group).
Итак, у нас в двух разных директориях имеются два проекта: DLL (для определенности пусть этот проект с именем geta.dpr находится в директории Z:\111111\Иванов\DLL) и вызывающая программа (project1.dpr в директории Z:\111111\Иванов\Caller). Объединим их в группу. Меню File(New(Other, выбираем вариант Project Group [image: image3.png]


. Сразу сохраним группу (File(Save As) в директории Z:\111111\Иванов под именем, например, group.bpg. В дальнейшем при запуске Delphi будем открывать именно файл group.bpg. Тогда на диске все наши файлы и проекты образуют следующую структуру (Рис.  2): 
[image: image4.wmf]Z:

111111

Иванов

DLL

Caller

Group.bpg

Geta.dpr

Project1.dpr

Geta.dll


Рис.  2 – Структура директорий и файлов в группе проектов 

(файлы выделены прямоугольниками).

Рис.  3 – Окно Менеджера проектов.
При создании группы на кране отобразится окно Менеджера проектов (Project Manager, Рис.  3). Это окон нам будет нужно часто, поэтому запомним, что оно вызывается клавишами Ctrl+Alt+F11. Добавим в группу два существующих проекта. Щелкаем правой кнопкой мыши в окне менеджера проектов и из контекстного меню выбираем пункт Add Existing Project. В открывшемся окне выбора файла указываем файл Project1.dpr в директории Caller. Проект Project1 загружается в Delphi. Снова вызываем окно менеджера и аналогично добавляем файл geta.dpr из директории DLL. 
Поскольку в один момент времени программист может работать только с одним проектом – или библиотекой, или вызывающей программой – менеджер проектов позволяет переключаться между ними, назначая активный проект. Он выделен жирным шрифтом в окне менеджера. Для выбора можно воспользоваться раскрывающимся списком или просто дважды щелкнуть по названию проекта. Вся работа (компиляция, отладка и пр.) идет только с активным проектом! 

Еще одно удобное свойство менеджера проектов – быстрый переход к любому файлу в любом проекте. Например, вы хотите открыть файл unit1.pas из проекта project1.dpr. Найдите его в дереве, отображаемом в окне менеджера, щелкните по нему правой кнопкой мыши и выберите пункт меню Open (для файлов *.pas) или View Source (для файлов *.dpr). 
Связывание проектов на этом еще далеко не завершено. Во-первых, при компиляции библиотеки файл geta.dll создается в директории Z:\111111\Иванов\DLL, а для его использования вызывающей программой он должен попасть в директорию Z:\111111\Иванов\Caller. Каждый раз перетаскивать его туда ручками – не лучшее решение. Во-вторых, так и не решена задача отладки, так как два проекта по-прежнему ничего не знают друг о друге. 

Для решения первой проблемы активируем проект DLL и идем в пункт меню Project(Options. На закладке Directories/Conditionals (Рис.  4) в поле Output Directory указываем Z:\111111\Иванов\Caller. Теперь при компиляции файл библиотеки geta.dll будет создаваться не в текущей директории, а в указанной. 

Рис.  4 – Окон опций проекта.

Рис.  5 – Задание вызывающей программы.
Остается указать, какой exe-файл будет вызывать нашу DLL (он называется host application). Пункт меню Run(Parameters, в поле Host Application указываем файл Z:\111111\Иванов\Caller\Project1.exe.
Вот теперь все! Можно запустить на выполнение обычным образом проект с библиотекой. При этом сначала запустится вызывающее приложение, а потом – библиотека. Можно ставить точки прерывания и отлаживать программу. Аналогично, если активировать проект project1, при пошаговой отладке можно будет "зайти" в проект DLL.
5. Хранение форм в DLL-библиотеках

 То, что вы храните достаточно долго, можно выбросить. 
Как только вы что-то выбросите, оно вам понадобится.
Из законов Мэрфи

В DLL-библиотеке можно хранить не только процедуры и функции, но и экранные формы с визуальными компонентами. Это позволяет в полной мере реализовать модульную структуру программы, когда к головному модулю подсоединяются DLL, реализующие те или иные возможности. По такому принципу строятся многие расширяемые системы, к которым можно добавлять все новые и новые модули. 
Добавим форму в проект geta.dll. Активируем его и создадим новую форму (команда меню File(New(Form). Сохраним созданный файл в директории Z:\111111\Иванов\DLL под произвольным именем, скажем, main.pas. Переименуем созданную форму (свойство Name) в DLLForm. В файле main.pas в разделе IMPLEMENTATION опишем процедуру:
PROCEDURE ShowModalForm;

BEGIN

 DLLForm := TDllForm.Create(Application);

 DLLForm.ShowModal;

 DLLForm.Free
END;
Данная процедура выполняет трюк барона Мюнхгаузена с поднятием самого себя за волосы: она создает форму DLLForm (метод Create), отображает ее  (метод ShowModal) и после закрытия пользователем выгружает из памяти (метод Free). 
В раздел описаний надо добавить заголовок процедуры ShowModalForm, чтобы она была доступна из файла geta.dll:

type

  TDLLForm = class(TForm)

  private

    { Private declarations }

  public

    { Public declarations }

  end;

procedure ShowModalForm;

var

  DLLForm: TDLLForm;

implementation

…

В файле geta.dll в раздел EXPORTS добавляем имя экспортируемой процедуры ShowModalForm: 
EXPORTS

 ShowModalForm, GetArea;

Библиотека с формой готова. Кстати, на форму можно поместить любые компоненты. Для вызова формы из библиотеки применим применим статическое связывание. В проекте project1.dpr поместим на форму еще одну кнопку и в ее обработчике запишем:

procedure TForm1.Button2Click(Sender: TObject);

begin

 ShowModalForm

end;

А после оператора IMPLEMENTATION запишем:

procedure ShowModalForm; FAR; EXTERNAL 'geta';
Готово! При нажатии на кнопку в программе Project1.exe на экране появится вторая форма, хранящаяся в библиотеке geta.dll. 

